Appendix I: Using Summation Notation With GAMS

Summation notation is difficult for some students to use and follow. Here we present notes on the mechanics of summation notation usage and some rules for proper use. This discussion is cast within the GAMS framework with presentation equivalents of common summation expressions and error messages caused by improper summation. All of the GAMS statements used herein are shown in Table 1 and are in file NOTATION.

AI.1 Summation Mechanics

Summation notation is a short hand way of expressing sums of algebraic terms involving subscripted items. In order to cover the mechanics of summation notation it is useful to have a set of subscripted items and associated numerical values. Thus, let us define some data

$x_1 = 1 \quad y_{11} = 2 \quad y_{12} = 3$
$x_2 = 2 \quad y_{21} = 4 \quad y_{22} = 1$
$x_3 = 3 \quad y_{31} = 1 \quad y_{32} = 4.$

Now let us define a variety of summation expressions.

AI.1.1 Sum of an Item.

Suppose we wished to sum all values of x. This would be written as

$$\sum_{i=1}^{3} x_i = x_1 + x_2 + x_3 = 1 + 2 + 3 = 6$$

or in GAMS

```
SUM1 = SUM(I, X(I));
```

For short hand purposes if i was to be summed over all possible values, we would write this as

$$\sum_{i} x_i .$$

We might also express a sum as follows which indicates all of the i are summed over except i=3

$$\sum_{\substack{i \\ i\_3}} x_i = 3 .$$

In GAMS, this is more difficult to express where one has to write a conditional ($) operation or define a subset as follows

```
SUM1 = SUM(I$(ORD(I.NE.3)), X(I));
```

or

```
SET SUBSETI(I)      /1, 2/;
SUM1 = SUM(SUBSETI, X(SUBSETI(I)));
```

AI.1.2 Multiple Sums

Sums over two indices consider all combinations of those items

$$\sum_i \sum_j y_{ij} = y_{11} + y_{12} + y_{21} + y_{22} + y_{31} + y_{32} = 15.$$

The equivalent GAMS expression is

```
SUM2 = SUM((I,J), Y(I,J));
```

AI.1.3 Sum of Two Items

Suppose we wished to sum over two items completely where they shared a subscript

$$\sum_{i=1}^{3} (x_i + \sum_{j=1}^{2} y_{ij}) = \sum_i (x_i + \sum_j y_{ij}) = \sum_i x_i + \sum_i \sum_j y_{ij} = x_1 + y_{11} + y_{12} + x_2 + y_{21} + y_{22} + x_3 + y_{31} + y_{32} = 21.$$

The equivalent GAMS expression is as follows

```
SUM3 = SUM(I, X(I)+SUM(J, Y(I, J)));
```

or

```
SUM3 = SUM(I, X(I)) + SUM((I,J), Y(I,J));
```

On the other hand, if we wished to sum the results only for the i^{th} element and call it A_i then

$A_i = x_i + \sum_j y_{ij} = x_i + y_i 1 + y_i 2$ or in GAMS

```
A(I) = X(I) + SUM(J, Y(I,J));
```

which would yield a vector [6 , 7 , 8] of results.

Sums over common subscripts can be collapsed or taken apart

or

$$\sum_i (x_i + z_i) = \sum_i x_i + \sum_i z_i$$

```
SUM4 = SUM(I, X(I) +Z(I));
```

or

```
SUM4 = SUM(I, X(I)) + SUM(I, Z(I));
```

AI.2 Summation Notation Rules

Certain rules apply when writing summation notation equations. The applicable rules depend on whether the final result is an unsubscripted scalar or a subscripted family of results determined by multiple equations.

AI.2.1 For a Scalar Equation

$$B1 = \sum_i \sum_j \sum_k p_{ijk} + \sum_m \sum_n qsubmn.$$

All subscripts must be dealt with in each term. Thus, it is proper to define the equation

However, the following equations are wrong

$$B2 = p_{ijk} + q_{mn}$$

$$B3 = \sum_j \sum_i p_{ijk} + \sum_m \sum_n q_{mn}.$$

In the case of the first equation, the result would really have the subscripts i,j,k,m,n,

while the second equation result would have to have a k subscript on B3 or a sum over k to be proper.

Equivalent GAMS commands for the above equation expressions are

```
EQB1..  B1 =E= SUM((I,J,K),P(I,J,K)) + SUM((M,N), Q(M,N));
EQB2..  B2 =E= P(I,J,K) + Q(M,N);
EQB3..  B3 =E= SUM((I,J), P(I,J,K)) + SUM((M,N), Q(M,N));
```

Here, the first equation expression is correct, while the last two equation expressions are incorrect. If you run GAMS with the above commands, you would encounter GAMS error messages $149 which says

"UNCONTROLLED SET ENTERED AS CONSTANT" meaning that you have not somehow dealt with all the subscripts in the equation.

AI.2.2. For a Family of Equations

Several rules apply when one is working with a family of equations.

1. The members of the family must be specified with an indication of the subscripts which define each equation. This is done by indicating all the conditions for which the equations exist in a "for" condition. For example, suppose we define an equation which sets all C's equal to 2. This is done by saying

$$C_i = 2 \quad \text{for all } i \quad \text{or} \quad C_i = 2 \quad \text{for } i = 1,2, \ldots n.$$

Similarly, if we wish to set a 2 dimensional variable equal to a constant, we would state

$$D_{ij} = 2 \text{ for all } i \text{ and } j,$$

while stating that for each row of the matrix E_{ij} we have the same values F_i is defined by

$$E1_{ij} = F_i \qquad \text{for all } i \text{ and } j.$$

The equivalent GAMS commands for the above expressions are

```
EQUATIONS
        EQC(I)          EQUATION C
        EQD(I,J)        EQUATION D
        EQE1(I,J)       EQUATION E1;
EQC(I).. C(I) =E= 2;
EQD(I,J)..      D(I,J) =E= 2;
EQE1(I,J)..     E1(I,J) =E= F(I);
```

On the other hand, it is wrong to state

$$E2_{ij} = 2$$

without conditions on i and j. The equivalent GAMS commands for the above incorrect expressions are

```
EQUATION
        EQE2            EQUATION E2;
        EQE2..          E2(I,J) =E= 2;
```

Here you would get error message $149 which says "UNCONTROLLED SET

ENTERED AS CONSTANT."

2. When writing an equation with a for statement, all subscripts which are not in the for

$$\sum_j \sum_k p_{ijk} = G1_i \qquad \text{for all } i$$

$$\sum_k p_{ijk} = H1_i \qquad \text{for all } i \text{ and } j$$

statement must be summed over. Consequently, it is proper to write

but improper to write

$$p_{ijk} = G2_i \qquad \text{for all } i$$

$$\sum_k p_{ijk} = H2_i \qquad \text{for all } i.$$

The equivalent GAMS commands for the above equations are

```
EQUATIONS
        EQG1(I)EQUATION G1
        EQH1(I,J)       EQUATION H1
        EQG2(I)EQUATION G2
        EQH2(I)EQUATION H2;
EQG1(I)..       G1(I)   =E= SUM((J,K), P(I,J,K));
EQH1(I,J)..     H1(I,J) =E= SUM(K, P(I,J,K));
EQG2(I)..       G2(I)   =E= P(I,J,K);
EQH2(I)..       H2(I)   =E= SUM(K, P(I,J,K));
```

in which the first two equations are correct, while the last two equations are wrong and error messages $149 "UNCONTROLLED SET ENTERED AS CONSTANT" would again be realized.

3. In any term of an equation, the result after executing the mathematical operations in that term must be of a dimension less than or equal to the family definition in the for statement. For example, it is proper to write

$$\sum_j \sum_k p_{ijk} = L1 \qquad \text{for all i}$$

$$\sum_j \sum_k r_{ijkm} + \sum_j s_{ijm} = N_{im} \qquad \text{for all i and m}$$

but wrong to write

$$p_{ijk} = L2 \qquad \text{for all i.}$$

Thus, for the following expressions, the first two equations are appropriate but the last equation would give you error message $149 "UNCONTROLLED SET ENTERED AS CONSTANT."

```
EQUATION
EQLI(I)...      LI(I) =E=       SUM((J,K), P(IJK));
EQN(I,M)...     N(I,M) =E=      SUM((J,K), R(I,J,K,M))
                                + SUM (J,S(I,J,M));
EQL2(I)...      L2 =E=  P(I,J,K);
```

4. When the dimension is less than the family definition this implies the same term appears in multiple equations. For example, in the equation

$$2 + \sum_j \sum_k p_{ijk} + \sum_j s_{ijm} = O_{im} \qquad \text{for all i and m,}$$

the 2 term appears in every equation and the sum involving p is common when m varies.

Equivalent GAMS commands are as follows

```
EQUATION
    EQO(I,M)     EQUATION O;
EQO(I,M)..    2 + SUM((J,K), P(I,J,K)) + SUM(J, S(I,J,M)) =E= O(I,M);
```

5. In an equation you can never sum over the parameter that determines the family of equations. It is certainly wrong to write

$$\sum_k \sum_j \sum_i p_{ijk} = W_i \qquad \text{for all i.}$$

Or, equivalently, the following expressions are wrong and will result in error

message $125 which says "SET IS UNDER CONTROL ALREADY."

```
EQW(I)...        W(I) =E=        SUM(I,J,K), P(I,J,K));
```

AI.3 Defining Subscripts

In setting up a set of equations and variables use the following principles. Define a subscript for each physical phenomena set which has multiple members, i.e.,

Let i denote production processes of which there are I
j denote locations of which there are J
k denote products of which there are K
m denote sales locations of which there are M.

Equivalent GAMS commands are

```
SET     I       /1*20/
        J       /1*30/
        K       /1*5/
        M       /CHICAGO, BOSTON/;
```

Define different subscripts when you are either considering subsets of the subscript set or different physical phenomena.

AI.4 Defining and Using Variables

1. Define a unique symbol with a subscript for each manipulatable item.

   For example:

   p_{ijk} = production using process i at location j while producing good k.

   Or, equivalently,

   ```
   PARAMETER    P(I,J,K)
   ```

   or

   ```
   PARAMETER    PRODUCTION(PROCESS, LOCATION, GOOD)
   ```

   Here, for documentation purposes, the second expression is preferred.

2. Make sure that variable has the same subscript in each place it occurs.

Thus it is proper to write

$$\text{Max} \quad \sum_i \sum_j \sum_k t_{ijk}$$

$$\sum_i \sum_j t_{ijk} = 3 \qquad \text{for all k}$$

but wrong to write

$$\text{Max} \quad \sum_i \sum_j t_{ij}$$

$$\sum_i \sum_j t_{ijk} = 3 \qquad \text{for all k}$$

$$t_{ijk} \ \exists \ 0.$$

The second model would cause error message $148 indicating "DIMENSION DIFFERENT."

3. The authors feel it is a bad practice to define different items with the same symbol but varying subscripts. We think you should never use the same symbol for two different items as follows

u_{ij} = amount of tires transported from i to j and

u_{kj} = amount of chickens transported from k to j.

GAMS would not permit this, giving error $150 "Symbolic Equations Redefined."

AI.5 Equations

Modelers should carefully identify the conditions under which each equation exists and use subscripts to identify those conditions. We do not think modelers should try to overly compact the families of equations. For example, it is OK to define

$$\sum_j a_{ij} x_j \# b_i$$

for all i, where a_{ij} is use of water by period and labor by period, where i denotes water periods and labor periods and b_i simultaneously contains water and labor availability by period. But we find it is better to define

$$\sum_j d_{ij} x_j \# e_i$$

$$\sum_j f_{ij} x_j \# h_i$$

where i denotes period,

d_{ij} denotes water use and e_i water availability,

f_{ij} denotes labor use and h_i labor availability.

AI.6 Cautions and Extensions

1. Be careful when you sum over terms which do not contain the subscript you are summing over. This is equivalent to multiplying a term by the number of items in the sum.

$$\sum_{j=1}^{N} x_i = Nx_i$$

$$\sum_{j=1}^{3} X_2 = 3(2) = 6$$

Or, in GAMS

```
SUM5A = SUM(J, X("2"));
```

2. Be careful when you have a term in a family of equations which is of a lesser dimension than the family, this term will occur in each equation. For example, the expression

$$\sum_j x_j = z_i \qquad \text{for } i = 1,2,3$$

implies that simultaneously

$$\sum_j x_j = z_1$$

$$\sum_j x_j = z_2$$

$$\sum_j x_j = z_3.$$

3. The same rules as outlined above apply to product cases

$$\prod_{i=1}^{3} x_i = x_1 * x_2 * x_3.$$

Or, equivalently,

```
PRODUCTX = PROD(I, X(I));
```

4. The following relationships also hold for summation

a. $\sum_i K\, x_i = K \sum_i x_i$

b. $\sum_{i=1}^{n} KP = K \sum_{i=1}^{n} P = K\, n\, P$

c. $\sum_i \sum_j (v_{ij} + y_{ij}) = \sum_i \sum_j v_{ij} + \sum_i \sum_j y_{ij}$

d. $\sum_i \sum_j (x_i + y_{ij}) = n \sum_i x_i + \sum_i \sum_j y_{ij}$ when $j = 1,2,\ldots n$

Table 1. Sample GAMS Commands for Summation Notation Expressions

```
 1  ************************************************
 2  ** THIS FILE CONTAINS GAMS EXAMPLES IN SUPPORT **
 3  ** OF THE NOTES USING THE SUMMATION NOTATION   **
 4  ************************************************
 5
 6  SETS
 7      I /1*3/
 8      J /1*2/
 9      K /1*2/
10      M /1*2/
11      N /1*3/
12
13  PARAMETERS
14
15      X(I) /1 1,2 2,3 3/
16      Z(I) /1 2,2 4,3 6/
17
18  TABLE Y(I,J)
19
20             1    2
21    1        2    3
22    2        4    1
23    3        1    4;
24
25  TABLE V(I,J)
26
27             1    2
28    1        2    3
29    2        4    1
30    3        1    4;
31
32  TABLE P(I, J, K)
33
34         1.1  1.2  2.1  2.2
35    1      1    3    5    7
36    2      2    4    6    8
37    3      1    2    3    4 ;
38
39  TABLE Q(M, N)
40
41         1   2   3
42    1    1   5  10
43    2   10   5   1;
44
45  ***************************
46  ** AI.1.1 SUM OF AN ITEM **
47  ***************************
48
49  PARAMETER
50   SUM1       SUM OF AN ITEM;
51   SUM1     = SUM(I, X(I));
52   DISPLAY SUM1;
53
54  ************************
55  ** AI.1.2 MULTIPLE SUMS **
56  ************************
57
58  PARAMETER
59   SUM2       MULTIPLE SUMS;
60   SUM2     = SUM((I,J), Y(I,J));
61   DISPLAY SUM2;
```

Table 1. Sample GAMS Commands for Summation Notation Expressions (continued)

```
 62
 63  ****************************
 64  ** AI.1.3 SUM OF TWO ITEMS **
 65  ****************************
 66
 67  PARAMETERS
 68   SUM3A     SUM OF TWO ITEMS-1
 69   SUM3B     SUM OF TWO ITEMS-1
 70    A(I)       SUM OF TWO ITEMS-2
 71    SUM4A      SUM OF TWO ITEMS-3
 72    SUM4B      SUM OF TWO ITEMS-3;
 73    SUM3A     = SUM(I, X(I)+SUM(J, Y(I, J)));
 74    SUM3B     = SUM(I, X(I)) + SUM ((I,J), Y(I,J));
 75    A(I)      = X(I) + SUM(J, Y(I,J));
 76    SUM4A     = SUM(I, X(I)+Z(I));
 77    SUM4B     = SUM(I, X(I)) + SUM(I, Z(I));
 78    DISPLAY SUM3A, SUM3B, A, SUM4A, SUM4B;
 79
 80  *********************************
 81  ** AI.2.1 FOR A SCALER EQUATION **
 82  *********************************
 83
 84  PARAMETERS
 85   B1   SUM FOR A SCALER EQUATION-1;
 86   B1 = SUM((I,J,K), P(I,J,K)) + SUM((M,N), Q(M,N));
 87   DISPLAY B1;
 88
 89  * $ONTEXT
 90  * THE FOLLOWING SUMMATION NOTATIONS ARE INCORRECT
 91  * IF YOU TURN THESE COMMANDS ON, YOU WILL ENCOUNTER
 92  * ERROR MESSAGES
 93  * PARAMETERS
 94  *  B2   SUM FOR A SCALER EQUATION-2
 95  *  B3   SUM FOR A SCALER EQUATION-3;
 96  *  B2 = P(I,J,K) + Q(M,N);
 97  *  B3 = SUM((I,J), P(I,J,K)) + SUM((M,N), Q(M,N));
 98  *  DISPLAY B2, B3;
 99  * $OFFTEXT
100
101  ***************************************
102  ** A.I.2.2 FOR A FAMILY OF EQUATIONS **
103  ***************************************
104
105  VARIABLES      C(I), D(I,J), E1(I,J), F(J);
106  EQUATIONS
107                 EQC(I)    EQUATION C
108                 EQD(I,J)  EQUATION D
109                 EQE1(I,J) EQUATION E1;
110       EQC(I)..        C(I) =E= 2;
111       EQD(I,J)..      D(I,J) =E= 2;
112       EQE1(I,J)..     E1(I,J) =E= F(J);
113
114  * $ONTEXT
115  * THE FOLLOWING EXPRESSION IS INCORRECT
116  * ERROR MESSAGES WILL BE ENCOUNTERED
```

```
 117  * VARIABLES E2(I,J);
```

Table 1. Sample GAMS Commands for Summation Notation Expressions (continued)

```
 118  * EQUATION
 119  *         EQE2      EQUATION E2;
 120  *         EQE2..    E2(I,J) =E= 2;
 121  * $OFFTEXT
 122
 123  VARIABLES G1(I), H1(I,J);
 124  EQUATIONS
 125           EQG1(I)   EQUATION G1
 126           EQH1(I,J) EQUATION H1;
 127      EQG1(I)..      G1(I) =E= SUM((J,K), P(I,J,K));
 128      EQH1(I,J)..    H1(I,J) =E= SUM(K, P(I,J,K));
 129
 130  * $ONTEXT
 131  * THE FOLLOWING EXPRESSIONS ARE INCORRECT
 132  * ERROR MESSAGES WILL BE ENCOUNTERED
 133  * VARIABLES G2(I), H2(I);
 134  * EQUATIONS
 135  *          EQG2(I)   EQUATION G2
 136  *          EQH2(I)   EQUATION H2;
 137  *     EQG2(I)..      G2(I) =E= P(I,J,K);
 138  *     EQH2(I)..      H2(I) =E= SUM(K, P(I,J,K));
 139  * $OFFTEXT
 140
 141  VARIABLES L1(I), U(I,M), R(I,J,K,M), S(I,J,M);
 142  EQUATIONS
 143            EQL1(I)   EQUATION L1
 144            EQN(I,M)  EQUATION N;
 145       EQL1(I)..      L1(I) =E= SUM((J,K), P(I,J,K));
 146       EQN(I,M)..     U(I,M) =E= SUM((J,K),R(I,J,K,M)) + SUM(J, S(I,J,M));
 147
 148  * $ONTEXT
 149  * THE FOLLOWING EXPRESSIONS ARE INCORRECT
 150  * ERROR MESSAGES WILL BE ENCOUNTERED
 151  * VARIABLES L2;
 152  * EQUATIONS
 153  *          EQL2(I)   EQUATION L2;
 154  *     EQL2(I)..      L2 =E= P(I,J,K);
 155  * OFFTEXT
 156
 157  VARIABLE O(I,M);
 158  EQUATION
 159          EQO(I,M) EQUATION O;
 160     EQO(I,M)..     2 + SUM((J,K), P(I,J,K)) + SUM(J, S(I,J,M)) =E=
O(I,M);
 161
 162
 163  * $ONTEXT
 164  * THE FOLLOWING EXPRESSION IS INCORRECT
 165  * GAMS ERROR MESSAGES WILL BE ENCOUNTERED
 166  * VARIABLE W(I);
 167  * EQUATION
 168  *         EQW(I)    EQUATION W;
 169  *     EQW(I)..  W(I) =E= SUM((I,J,K), P(I,J,K));
 170  * $OFFTEXT
 171
```

Table 1. Sample GAMS Commands for Summation Notation Expressions (continued)

```
172  ************************************
173  ** AI.4 DEFINING AND USING VARIABLES **
174  ************************************
175
176  VARIABLES
177    OBJ1        OBJECTIVE FUNCTION VALUE
178    T(I,J,K)   DECISION VARIABLE;
179  EQUATIONS
180    OBJFUNC1    OBJECTIVE FUNCTION
181    CONST(K)  CONSTRAINT;
182    OBJFUNC1..    OBJ1 =E= SUM((I,J,K), T(I,J,K));
183    CONST(K)..  SUM((I,J), T(I,J,K)) =E= 3;
184  MODEL EXAMPLE1 /ALL/;
185  SOLVE EXAMPLE1 USING LP MAXIMIZING OBJ1;
186  DISPLAY T.L;
187
188  * $ONTEXT
189  * THE FOLLOWING COMMANDS ARE INCORRECT
190  * THEY WILL RESULT IN ERROR MESSAGES
191  * VARIABLES
192  *   OBJ2        OBJECTIVE FUNCTION VALUE
193  *   TT(I,J,K)   DECISION VARAIBLE;
194  * POSITIVE VARIABLE TT;
195  * EQUATIONS
196  *   OBJFUNC2    OBJECTIVE FUNCTION
197  *   CONSTT(K)   CONSTRAINT;
198  *   OBJFUNC2..   OBJ2 =E= SUM((I,J), TT(I,J));
199  *   CONSTT(K)..  SUM((I,J), TT(I,J,K)) =E= 3;
200  * MODEL EXAMPLE2 /ALL/;
201  * SOLVE EXAMPLE2 USING LP MAXIMIZING OBJ2;
202  * DISPLAY TT.L;
203  * $OFFTEXT
204
205  ********************************
206  ** AI.6 CAUTIONS AND EXTENSIONS **
207  ********************************
208
209  PARAMETER
210    SUM5A  CAUTIONS AND EXTENSIONS-1;
211    SUM5A = SUM(J, X("2"));
212    DISPLAY SUM5A;
213
214  PARAMETER
215    PRODUCT6 CAUTIONS AND EXTENSIONS-2;
216    PRODUCT6 = PROD(I, X(I));
217    DISPLAY PRODUCT6;
218
219  PARAMETERS
220    SUM7A    CAUTIONS AND EXTENSIONS-3
221    SUM7B    CAUTIONS AND EXTENSIONS-3
222    SUM8A    CAUTIONS AND EXTENSIONS-4
223    SUM8B    CAUTIONS AND EXTENSIONS-4
224    SUM8C    CAUTIONS AND EXTENSIONS-4
225    SUM9A    CAUTIONS AND EXTENSIONS-5
```

```
226    SUM9B    CAUTIONS AND EXTENSIONS-5
```

Table 1. Sample GAMS Commands for Summation Notation Expressions (continued)

```
227    SUM10A   CAUTIONS AND EXTENSIONS-6
228    SUM10B   CAUTIONS AND EXTENSIONS-6;
229    SUM7A =  SUM(I, 5*X(I));
230    SUM7B =  5*SUM(I, X(I));
231    SUM8A =  SUM(I, 5*10);
232    SUM8B =  5*SUM(I, 10);
233    SUM8C =  5*3*10;
234    SUM9A = SUM((I,J), V(I,J)+Y(I,J));
235    SUM9B = SUM((I,J), V(I,J)) + SUM((I,J), Y(I,J));
236    SUM10A = SUM((I,J), X(I)+Y(I,J));
237    SUM10B = 2*SUM(I, X(I)) + SUM((I,J), Y(I,J));
238    DISPLAY SUM7A, SUM7B, SUM8A, SUM8B, SUM8C,
239            SUM9A, SUM9B, SUM10A, SUM10B;
```